

Reading and Writing Arithmetic: Automating Ring Equalities in Agda

ANONYMOUS AUTHOR(S)

```

lemma : ∀ x y → x + y * 1 + 3 ≈ 2 + 1 + y + x

lemma x y = begin
  x + y * 1 + 3 ≈⟨ refl ⟨ +-cong ⟩ *-identityr y ⟨ +-cong ⟩ refl {3} ⟩
  x + y + 3    ≈⟨ +-comm x y ⟨ +-cong ⟩ refl ⟩
  y + x + 3    ≈⟨ +-comm (y + x) 3 ⟩
  3 + (y + x)  ≈⟨ sym (+-assoc 3 y x) ⟩
  2 + 1 + y + x ■

```

(a) A Tedious Proof

(b) Our Solver

Fig. 1. Comparison Between A Manual Proof and The Automated Solver

We present a new library which automates the construction of equivalence proofs between polynomials over commutative rings and semirings in the programming language Agda [Norell and Chapman 2008]. It is significantly faster than Agda’s existing solver. We use reflection to provide a simple interface to the solver, and demonstrate a novel use of the constructed relations: step-by-step solutions.

Additional Key Words and Phrases: proof automation, equivalence, proof by reflection, step-by-step solutions

1 INTRODUCTION

Doing mathematics in dependently-typed programming languages like Agda has a reputation for being tedious, awkward, and difficult. Even simple arithmetic identities like the one in Fig. 1 require fussy proofs (Fig. 1a).

This need not be the case! With some carefully-designed tools, mathematics in Agda can be easy, friendly, and fun. This work describes one such tool: an Agda library which automates the construction proofs like Fig. 1a, making them as easy as Fig. 1b.

As you might expect, our solver comes accompanied by a formal proof of correctness. Beyond that, though, we also satisfied the following requirements:

Friendliness and Ease of Use Proofs like the one in Fig. 1a aren’t just boring; they’re *difficult*.

The programmer needs to remember the particular syntax for each step (“is it `+-comm` or `+-commutative?`”), and often they have to put up with poor error messages.

We believe this kind of difficulty is why Agda’s current ring solver [Danielsson 2018] enjoys little widespread use. Its interface (Fig. 2) is almost as verbose as the manual proof, and it requires programmers to learn another syntax specific to the solver.

Our solver strives to be as easy to use as possible: the high-level interface is simple (Fig. 1b), we don’t require anything of the user other than an implementation of one of the supported algebras, and effort is made to generate useful error messages.

```
lemma = +*-Solver.solve 2 (λ x y → x := y : * con 1 := con 3 := con 2 := con 1 := y := x) refl
```

Fig. 2. The Old Solver




Fig. 3. The Reflexive Proof Process

Performance Typechecking dependently-typed code is a costly task. Automated solvers like the one presented here can greatly exacerbate this cost: in our experience, it wasn't uncommon for Agda's current ring solver to spend upwards of 10 minutes proving a single identity. In practice, this means two things: firstly, large libraries for formalising mathematics (like Meshveliani [2018]) can potentially take hours to typecheck (by which time the programmer has understandably begun to reconsider the whole notion of mathematics on a computer); secondly, certain identities can simply take too long to typecheck, effectively making them "unprovable" in Agda altogether!

The kind of solver we provide here is based on Coq's [Team 2018] ring tactic, described in Grégoire and Mahboubi [2005]. While we were able to apply the same optimisations that were applied in that paper, we found that the most significant performance improvements came from a different, and somewhat surprising part of the program. In terms of both practical use and theoretical bounds, our solver is significantly faster than Agda's current solver.

Educational Features While our solver comes with the benefit of formal correctness, it's still playing catch-up to other less-rigorous computer algebra systems in terms of features. These features have driven systems like Wolfram|Alpha [Wolfram Research, Inc. 2019] to widespread popularity among (for instance) students learning mathematics.

We will take just one of those features ("pedagogical", or step-by-step solutions [The Development Team 2009]), and re-implement it in Agda using our solver. In doing so, we will formalise the concept, and explore some of the theory behind it.

2 OVERVIEW OF THE PROOF TECHNIQUE

There are a number of ways we can automate proofs in a dependently-typed programming language, including Prolog-like proof search [Kokke and Swierstra 2015], Cooper's algorithm over Presburger arithmetic [Allais 2011], etc. Here, we will use a reflexive technique [Boutin 1997] in combination with sparse Horner Normal Form. The high-level diagram of the proof strategy is presented in Fig. 3.

The identity we'll be working with is the lemma in Fig. 1: the left and right hand side of the equality are at the bottom of the diagram. Our objective is to link those two expressions up through repeated application of the ring axioms. We do this by converting both expressions to a normal form (seen at the top of the diagram), and then providing a proof that this conversion is correct according to the ring axioms (the `correct` function in the diagram). Finally, we link up all of these proofs, and if the two normal forms are definitionally equal, the entire thing will typecheck, and we will have proven the equality.

2.1 The `Expr` AST

Instead of manipulating the expressions directly, we will define an Abstract Syntax Tree (AST) to describe expressions which use the ring operators.

```
data Expr {ℓ} (A : Set ℓ) (n : ℕ) : Set ℓ where
  K   : A → Expr A n
  l   : Fin n → Expr A n
  _+_ : Expr A n → Expr A n → Expr A n
  _*_ : Expr A n → Expr A n → Expr A n
  _⊗_ : Expr A n → ℕ → Expr A n
  _⊖_ : Expr A n → Expr A n
```

Each of the ring operators has a corresponding constructor ($x + y = x \oplus y$, $x^y = x \otimes y$), constants are constructed with `K`, and variables are referred to by their de Bruijn index (so x becomes `l 0`). The ASTs for both expressions we want to prove can be seen on either side of Fig. 3.

From here, we can “evaluate” the AST in one of two ways: in a non-normalised way (`[_]_`), or in a normalising way (`[_]_↓`). This means that the goal of the `correct` function is to show equivalence between `[_]_` and `[_]_↓`.

Finally, we *don't* want to force users to construct the `Expr` AST themselves. This is where reflection comes in: it automates this construction (the path labelled `quoteTerm` in the diagram) from the goal type.

2.2 Almost Rings

While the stated domain of the solver is simply “commutative rings”, it turns out that we can be slightly more flexible than that if we pick our algebra carefully. As in Grégoire and Mahboubi [2005, section 5], we use an algebra called an *almost-ring*. It has the regular operations ($+$, $*$ (multiplication), $-$, 0 , and 1), such that the following equations hold:

$$0 + x = x \tag{1}$$

$$x + y = y + x \tag{2}$$

$$x + (y + z) = (x + y) + z \tag{3}$$

$$1 * x = x \tag{4}$$

$$x * y = y * x \tag{5}$$

$$x * (y * z) = (x * y) * z \tag{6}$$

$$(x + y) * z = x * z + y * z \tag{7}$$

$$0 * x = 0 \tag{8}$$

$$-(x * y) = -x * y \tag{9}$$

$$-(x + y) = -x + -y \tag{10}$$

The equations up to 8 represent a pretty standard definition of a (commutative) semiring. From there, though, things are different. The normal definition of a commutative ring would have (instead of 9 and 10) the following:

$$x + -x = 0 \quad (11)$$

However, by choosing these slightly more complex laws, we can admit types like `N` which don't have additive inverses. Instead, these types can simply supply the identity function for `-`, and then 9 and 10 will still hold.

A potential worry is that because we don't require $x + -x = 0$ axiomatically, it won't be provable in our system. Happily, this is not the case: as long as $1 + -1$ reduces to 0 in the coefficient set, the solver will verify the identity.

In the library, the algebra is represented by the `AlmostCommutativeRing` type, a record with fields for each of the ring axioms, defined over a user-supplied equivalence relation. Just as in Agda's current solver, we also ask for one extra function: a weakly decidable predicate to test if a constant is equal to zero.

`is-zero : $\forall x \rightarrow \text{Maybe } (0 \# \approx x)$`

This function is used to speed up some internal algorithms in the solver, but it isn't an essential component. By making it *weakly* decidable, we allow users to skip it (`is-zero = const nothing`) if their type doesn't support decidable equality, or provide it (and get the speedup) if it does.

3 THE INTERFACE

A decent interface is crucial if we want the solver to be broadly useful. We strove to make our interface as simple and *small* as possible. Aside from the `AlmostCommutativeRing` type described above, the user-facing portion of our library consists of just two macros: `solve` and `solveOver`. Each of these infer the goal from their context, and automatically construct the required machinery to prove the equality.

`solve` is demonstrated in Fig. 1b. It takes a single argument: an implementation of the algebra. `solveOver` is designed to be used in conjunction with manual proofs, so that a programmer can automate a "boring" section of a larger more complex proof. It is called like so:

```
lemma :  $\forall x y \rightarrow x + y * 1 + 3 \approx 2 + 1 + y + x$ 
lemma x y =
  begin
    x + y * 1 + 3  $\approx$  { +comm (x + y * 1) 3 }
    3 + (x + y * 1)  $\approx$  { solveOver (x :: y :: []) Nat.ring }
    3 + y + x  $\equiv$  { }
    2 + 1 + y + x ■
```

As well as the `AlmostCommutativeRing` implementation, this macro takes a list of free variables to use to compute the solution.

Because this interface is quite small, it's worth pointing out what's missing, or rather, what we *don't* require from the user:

- We don't ask the user to construct the `Expr` AST which represents their proof obligation. Compare this to Fig. 2: we had to write the type of the proof twice (once in the signature and again in the AST), and we had to learn the syntax for the solver's AST.

As well as being more verbose, this approach is less composable: every change to the proof type has to be accompanied by a corresponding change in the call to the solver. In contrast, the call to `solveOver` above effectively amounts to a demand for the compiler to “figure it out!” Any change to the expressions on either side will result in an *automatic* change to the proof constructed.

- We don’t ask the user to write any kind of “reflection logic” for their type. In other words, we don’t require a function which (for instance) recognises and parses the user’s type in the reflected AST, or a function which does the opposite, converting a concrete value into the AST that (when unquoted) would produce an expression equivalent to the quoted value.

This kind of logic is complex, and very difficult to get right. While some libraries can assist with the task [Norell 2018; van der Walt and Swierstra 2013] it is still not fully automatic.

Finally, despite the simplicity and ease-of-use described above, the solver is *not* specialised to a small number of types like `N` and so on. The whole library, including the reflection-based interface, will work with any type with an `AlmostCommutativeRing` instance.

3.1 Reflection

Agda has good support for reflection, which we will use to build our interface. Agda’s reflection system consists of three main parts:

Term The representation of Agda’s AST, retrievable via `quoteTerm`.

Name The representation of identifiers, retrievable via `quote`.

TC The type-checker monad, which includes scoping and environment information, can raise type errors, unify variables, or provide fresh names. Computations in the `TC` monad can be run with `unquote`.

While `quote`, `quoteTerm`, and `unquote` provide all the functionality we need, they’re somewhat low-level and noisy (syntactically speaking). Agda also provides a mechanism (which it calls “macros”) to package metaprogramming code so it looks like a normal function call (as in `solve`).

Reflection is obviously a powerful tool, but it has a reputation for being unsafe and error-prone. Agda’s reflection system doesn’t break type safety, but we *are* able to construct `Terms` which are ill-typed, which often result in confusing error-messages on the user’s end. Unfortunately, constructing ill-typed terms is quite easy to do: every quoted expression comes with heaps of contextual information, making the whole thing very fragile. Variables, for instance, are referred to by their de Bruijn indices, meaning that the same `Term` can break if it’s simply moved under a lambda.

All of that considered, we feel we managed to construct a reasonably robust reflection-based interface. In doing so, we came up with the following general guidelines for metaprogramming in Agda:

Supply the minimal amount of information. There were several instances where, in constructing a term, we were tempted to supply explicitly some argument that Agda usually infers. Universe levels were a common example. In general, though, this is a bad idea: AST manipulation is fragile and error-prone, so the chances that you’ll get some argument wrong are very high. Instead, you should *leverage* the compiler, relying on inference over direct metaprogramming as much as possible.

Don’t assume structure. A common pattern we used to try and find arguments to an n -ary function was to simply extract the last n visible arguments to the function. While in theory we might be able to statically know all of the implicit and explicit arguments that will be used at the call-site, it’s much simpler to ignore them, and try our best to be flexible. Remember, none of this is typed, so if something changes (like, say, a new universe level

in `AlmostCommutativeRing`), you'll get type errors where you call `solve`, not where it's implemented.

Ask for forgiveness, not permission. We could also here say “don't roll your own type-checker”. While it may seem good and fastidious to rigorously check the structure of the arguments given to a macro, often we found better results by assuming the argument was correct (where possible), and then carefully structuring the output in such a way to funnel a type error to the place where the input was incorrect. For instance, one section of the solver algorithm expects a proof that the two normal forms of the equations are the same. Here, we simply supply `refl`, assuming that they are, in fact, the same. When they're *not*, for instance, in the following type:

$$x + y * 1 + 3 \approx 2 + 1 + y + y$$

A call to `solve` will provide the reasonably helpful error message:

$$x \neq y \text{ of type } \mathbb{N}$$

Try and implement as much of the logic outside of reflection as possible Finally, after all of that, we advise minimising the amount of actual metaprogramming code in a program, and confining it to the edges as much as possible. With great power comes poor error messages, fragility, and a loss of first-class status. Therefore, If something can be done without reflection, *do it*, and use reflection as the glue to get from one standard representation to another.

To demonstrate some of these principles, we'll look at a couple of functions which are used in the core of the reflection mechanism. First, `toExpr`:

```
toExpr : Term → Term
toExpr (def (quote AlmostCommutativeRing._ + _) xs) = getBinOp (quote _ ⊕_) xs
toExpr (def (quote AlmostCommutativeRing._ * _) xs) = getBinOp (quote _ ⊗_) xs
toExpr (def (quote AlmostCommutativeRing._ ^ _) xs) = getExp xs
toExpr (def (quote AlmostCommutativeRing._ - _) xs) = getUnOp (quote _ ⊖_) xs
toExpr v@(var x _) with x ℕ.<? numVars
... | yes p = v
... | no ¬p = constExpr v
toExpr t = constExpr t
```

This function is called on the `Term` representing one side of the target equality. It converts it to the corresponding `Expr`. In other words, it performs the following transformation:

$$x + y * 1 + 3 \rightarrow I 0 \oplus I 1 \otimes K 1 \oplus K 3$$

It calls off to a number of helper functions (`getBinOp`, `getExp`, `getUnOp`) which extract the relevant arguments from an applied operator. We'll look at the one for exponentiation, whose definition is as follows:

```
getExp : List (Arg Term) → Term
getExp (x (::) y (::) []) = quote _ ⊗_ (con } 3 ...[::] toExpr x (::) y (::) []
getExp (x :: xs) = getExp xs
getExp _ = unknown
```

Now, we can see how the guidelines from above are applied.

First, `getExp` is flexible with regards to structure: it extracts the last two visible arguments (rather than the n^{th}) in the argument list (`(::)` is a pattern synonym which matches visible arguments), which means that it will still work even if `AlmostCommutativeRing` gets another hidden parameter.

Once it matches them, it applies the corresponding constructor for the `Expr` type (`quote _⊗_ { con }`), calling `toExpr` recursively on the first argument. We expect y to be a `N` here, but we don't check that ourselves: `⊗_` expects a `N` for its second argument anyway, so if there's a problem we'll get *exactly* the type error we want.

In between those two parts, we have `3 ...?::[]`. This applies three hidden arguments as “unknown”, i.e. asks Agda to infer them. We could guess them ourselves: the first is the universe level of the carrier type, the second is the carrier type, and the third is the number of variables in the expression; but, as explained before, it's better to trust the compiler to guess them than us.

Looking back to `toExpr`, we notice that the last line is a catch-all, which simply constructs a constant expression. This is the trick which lets us avoid any custom quotation machinery from the user. It's also more robust than asking for custom quotation machinery: if, for instance, there's a function call or something similar hidden in this case, quotation won't work. This solution, though, which just packages up the expression as-is, will have no trouble.

4 PERFORMANCE

Type-checking proof-heavy Agda code is notoriously slow, so the solver had to be carefully optimised to avoid being so slow as to be unusable. We'll start by first describing the unoptimised solver, and demonstrate how to improve its performance iteratively.

4.1 Horner Normal Form

The representation used in Agda's current ring solver (and the one we'll start out with here) is known as Horner Normal Form. A polynomial (more specifically, a monomial) in x is represented as a list of coefficients of increasing powers of x . As an example, the following polynomial:

$$3 + 2x^2 + 4x^5 + 2x^7 \quad (12)$$

Is represented by this list:

$$3 :: 0 :: 2 :: 0 :: 0 :: 4 :: 0 :: 2 :: []$$

Operations on these polynomials are similar to operations in positional number systems.

`_⊞_ : Poly → Poly → Poly`

`[] ⊞ ys = ys`

`(x :: xs) ⊞ [] = x :: xs`

`(x :: xs) ⊞ (y :: ys) = x + y :: xs ⊞ ys`

`_⊗_ : Poly → Poly → Poly`

`_⊗_ [] _ = []`

`_⊗_ (x :: xs) =`

`foldr (λ y ys → x * y :: map (λ y → y * y) xs ⊞ ys) []`

And finally, evaluation of the polynomial (given x) is a classic example of the `foldr` function.

`[_] : Poly → Carrier → Carrier`

`[xs] ρ = foldr (λ y ys → ρ * ys + y) 0# xs`

4.2 Sparse Encodings

Our first avenue for optimisation comes from Grégoire and Mahboubi [2005]. Notice that the encoding above is quite wasteful: it always stores an entry for each coefficient, even if it's zero. In practice, we're likely to often find long strings of zeroes (in expressions like x^{10}), meaning that our representation will contain long "gaps" between the coefficients we're actually interested in (non-zero ones).

To fix the problem we'll switch to a *sparse* encoding, by storing a "power index" with every coefficient.

```
Poly : Set c
Poly = List (Carrier × ℕ)
```

This will represent the size of the gap from the previous non-zero coefficient. Taking 12 again as an example, we would now represent it as follows:

```
(3, 0) :: (2, 1) :: (4, 2) :: (2, 1) :: []
```

Next, we turn our attention to the task of adding multiple variables. Luckily, there's an easy way to do it: nesting. Multivariate polynomials will be represented as "polynomials of polynomials", where each level of nesting corresponds to one variable. It's perhaps more clearly expressible in types:

```
Poly : ℕ → Set c
Poly zero = Carrier
Poly (suc n) = List (Poly n × ℕ)
```

Inductively speaking, a "polynomial" in 0 variables is simply a constant, whereas a polynomial in n variables is a list of coefficients, which are themselves polynomials in $n - 1$ variables.

Before running off to use this representation, though, we should notice that we have created another kind of "gap" which we should avoid with a sparse encoding. For a polynomial with n variables, we will always have n levels of nesting, even if the polynomial doesn't actually refer to all n variables. In the extreme case, representing the constant 6 in a polynomial of 3 variables looks like the following:

```
(((((6, 0) :: []), 0) :: []), 0) :: []
```

The solution is another index: this time an "injection" index. This represents "how many variables to skip over before you get to the interesting stuff". This particular optimisation is considerably more complex than the previous, though: the number of variables in a polynomial is a type-relevant piece of information, so any *manipulation* of that index will have to justify itself to the typechecker.

4.3 Hanging Indices

The problem is a common one: we have a piece of code that works efficiently, and we now want to make it "more typed", by adding more information to it, *without* changing the complexity class.

We found the following strategy to be useful: first, write the untyped version of the code, forgetting about the desired invariants as much as possible. Then, to add the extra type information, look for an inductive type which participates in the algorithm, and see if you can "hang" some new type indices off of it.

In our case, the injection index (distance to the next "interesting" polynomial) was simply stored as an \mathbb{N} , and the information we needed was the number of variables in the inner polynomial, and the number of variables in the outer. All of that is stored in the following proof of \leq :


```

data _≤_ (m : ℕ) : ℕ → Set where
  m≤m : m ≤ m
  ≤-s  : ∀ {n}
        → (m≤n : m ≤ n)
        → m ≤ suc n

```

A value of type $n \leq m$ mimics the inductive structure of the \mathbb{N} we were storing to represent the distance between n and m . We were able to take this analogy to the extreme: where we needed an equivalent of `Ordering`:

```

data Ordering : ℕ → ℕ → Set where
  less  : ∀ m k → Ordering m (suc (m + k))
  equal : ∀ m   → Ordering m m
  greater : ∀ m k → Ordering (suc (m + k)) m

```

We were able to construct one, with transitivity replacing addition.

```

data ≤-Ordering {n : ℕ} : ∀ {i j}
  → (i≤n : i ≤ n)
  → (j≤n : j ≤ n)
  → Set
  where
    ≤-lt : ∀ {i j-1}
      → (i≤j-1 : i ≤ j-1)
      → (j≤n : suc j-1 ≤ n)
      → ≤-Ordering (≤-trans (≤-s i≤j-1) j≤n)
    ≤-gt : ∀ {i-1 j}
      → (i≤n : suc i-1 ≤ n)
      → (j≤i-1 : j ≤ i-1)
      → ≤-Ordering i≤n (≤-trans (≤-s j≤i-1) i≤n)
    ≤-eq : ∀ {i}
      → (i≤i : i ≤ i)
      → ≤-Ordering i≤i i≤i
    ≤-compare : ∀ {i j n}
      → (x : i ≤ n)
      → (y : j ≤ n)
      → ≤-Ordering x y
    ≤-compare m≤m m≤m = ≤-eq m≤m
    ≤-compare m≤m (≤-s y) = ≤-gt m≤m y
    ≤-compare (≤-s x) m≤m = ≤-lt x m≤m
    ≤-compare (≤-s x) (≤-s y)
      with ≤-compare x y
    ... | ≤-lt i≤j-1 _ = ≤-lt i≤j-1 (≤-s y)
    ... | ≤-gt _ j≤i-1 = ≤-gt (≤-s x) j≤i-1
    ... | ≤-eq _      = ≤-eq (≤-s x)

```

4.4 Unification

So far, our optimisations have focused on the *operations* performed on the polynomial. Remember, though, the reflexive proof process has several steps: only one of them containing the operations (`[[_]]` in Fig. 3).

As it happens, we have now optimised these operations so much that they are no longer the bottleneck in the process. Surprisingly, the innocuous-looking `refl` now takes the bulk of the time! Typechecking this step involves unifying the two normalised expressions, a task which is quite expensive, with counterintuitive performance characteristics. So counterintuitive, in fact, that early versions of the solver, with all the optimisations from Grégoire and Mahboubi [2005] applied, was in many cases *slower* than the old, unoptimised solver!

In this section, we'll try and explain the problem and how we fixed it, and give general guidelines on how to write Agda code which typechecks quickly.

First, the good news. In the general case, unifying two expressions takes time proportional to the size of those expressions, so our hard-won optimisations do indeed help us.

Unfortunately, though, the “general case” isn’t really that general: Agda’s unification algorithm has a very important shortcut which we *must* make use of if we want our code to typecheck quickly: *syntactic equality*.

Because Agda is a dependently-typed language, types can contain functions, variables, and all sorts of complex expressions. One might expect that the unification algorithm should compute these expressions as far as it can, getting them to normal form, before it checks for any kind of equality. This would be disastrous for performance! Consider the following:

$$\text{sum } [1..100] \stackrel{?}{=} \text{sum } [1..100]$$

Running both computations here is an unnecessarily expensive task, and one which Agda does indeed avoid. Before the full unification algorithm, the typechecker does a quick pass to spot any syntactic equalities like the one above: if it sees one, it can avoid any more computation on that particular expression.

Taking advantage of that shortcut is key to achieving decent performance. With that in mind, there are two main strategies we’ll use to encourage syntactic equality:

4.4.1 Avoid Progress at all Costs. First, we will consider something which may seem inconsequential: the order of arguments to the evaluation functions.

$$[\![xs]\!]_l \rho = \text{foldr } (\lambda y \, ys \rightarrow \rho * ys + y) \, 0 \, xs \qquad [\![xs]\!]_r \rho = \text{foldr } (\lambda y \, ys \rightarrow y + ys * \rho) \, 0 \, xs$$

$[\![_]\!]_l$ is the definition we’ve been working with so far. Some readers might find $[\![_]\!]_r$ more natural, however. The reason is that it’s more productive: in lazy languages, the usual convention is that functions which take multiple arguments should scrutinise those arguments from left to right. The $*$ and $+$ functions (on \mathbb{N} , at any rate) follow that convention, meaning that $[\![_]\!]_r$ is able to make more progress without a concrete x . Taking the polynomial $x^2 + 2$ as an example:

$$\begin{array}{ll} [\![2 :: 0 :: 1 :: []]\!]_l x & \equiv \langle \rangle \\ x * (x * (x * 0 + 1) + 0) + 2 & \equiv \langle \rangle \\ x * (x * (x * 0 + 1) + 0) + 2 & \blacksquare \end{array} \qquad \begin{array}{ll} [\![2 :: 0 :: 1 :: []]\!]_r x & \equiv \langle \rangle \\ 2 + (0 + (1 + 0 * x) * x) * x & \equiv \langle \rangle \\ \text{succ } (\text{succ } ((x + 0) * x)) & \blacksquare \end{array}$$

In $[\![_]\!]_l$, progress is immediately blocked when we try to scrutinise x . In $[\![_]\!]_r$, since all of the constants are kept to the left, they’re scrutinised first, allowing us to perform much more normalisation before being blocked.

Surprisingly, this is exactly what you *don’t* want! Since both sides of the equation will be coming out of the same normalisation function, they should have similar structures, allowing for syntactic equality. The coefficients, though, will be computed during manipulations on the Horner normal form, and so may contain unevaluated expressions, which require normalisation. If we used $[\![_]\!]_r$ as our definition, then, the typechecker will likely hit an inequality very early on, and give up on syntactic equality. Even worse, because it’s able to make progress, it will likely do so, completely changing the structure of the expression and ruining any chance for later syntactic equalities!

The (counterintuitive) lesson learned is as follows: to speed up unification, keep things which are likely to be syntactically equal to the left, and *don’t* structure your functions to encourage progress. Simply swapping the arguments (as we do above) resulted in a performance improvement of several orders of magnitude.




Fig. 4. Comparison of the normal forms of equation 12

4.4.2 Avoid Constants. It's a good idea to avoid constant coefficient expressions in the normal form. This will reduce the size of your expression, which is helpful in general, but more importantly it will make it less likely that the typechecker will be able to run some normalisation on one of the ring operators.

Our sparse representation helps significantly in this case: by removing **0** from the generated expression, we significantly improve the chances of maintaining structural similarity between the two normal forms.

Another place we can cut down on constants is in identity elements in the base case for recursive functions. Take exponentiation, for example:

$$\begin{aligned}
 &_ \wedge _ : \text{Carrier} \rightarrow \mathbb{N} \rightarrow \text{Carrier} \\
 &x \wedge \text{zero} = 1 \\
 &x \wedge \text{suc } i = x^* (x \wedge i)
 \end{aligned}$$

We can avoid that **1** in the majority of cases by rewriting the function to have an extra base case:

$$\begin{aligned}
 &_ \wedge +1 : \text{Carrier} \rightarrow \mathbb{N} \rightarrow \text{Carrier} & _ \wedge _ : \text{Carrier} \rightarrow \mathbb{N} \rightarrow \text{Carrier} \\
 &x \wedge \text{zero} +1 = x & x \wedge \text{zero} = 1 \\
 &x \wedge \text{suc } i +1 = (x \wedge i +1)^* x & x \wedge \text{suc } i = x \wedge i +1
 \end{aligned}$$

In the library, we employ this idea extensively, avoiding unnecessary identities as much as we could. This has a significant effect on the size of the resulting normal form, but also ensures that normalisation stops exactly where we want it to, preserving the structure of the expressions as

much as is possible. This makes a significant difference to both size and syntactic similarity as can be seen in Fig. 4.

4.5 Verifying the Optimisations

Optimising extrinsically-verified code can be a tedious task. Every small change to an algorithm (like swapping the order of arguments to `*`) has the potential to invalidate your verification, occasionally forcing you to totally rewrite all of the proofs.

For our case, we were required to prove a ring homomorphism between Horner normal form and the carrier type. These proofs are long (about 1000 lines) and complex, and every new optimisation would require a whole new round of proof code, with very little reuse.

To remedy this problem, we took inspiration from Mu et al. [2009], and relied heavily on abstraction and folds to improve the reuse in proof code. In particular, we defined many operations as *metamorphisms* [Gibbons 2007]. So, instead of defining (say) negation over the polynomial type itself, we will define a metamorphism, which has the following type:

$$\begin{aligned} \text{Meta} &: \mathbb{N} \rightarrow \text{Set } c \\ \text{Meta } n &= \text{Poly } n \times \text{Coeff } n \star \rightarrow \text{Poly } n \times \text{Coeff } n \star \end{aligned}$$

And then we can run a metamorphism over a polynomial with some higher-order function.

From here, we can define the *semantics* of a metamorphism. As an example, here are the semantics of `mapR`, a simple morphism which behaves something like `map` on lists:

$$\begin{aligned} \text{poly-mapR} &: \forall \{n\} \rho \rho s \\ &\rightarrow ([f] : \text{Poly } n \rightarrow \text{Poly } n) \\ &\rightarrow (f : \text{Carrier} \rightarrow \text{Carrier}) \\ &\rightarrow (\forall x \ y \rightarrow x \star f \ y \approx f(x \star y)) \\ &\rightarrow (\forall x \ y \rightarrow f(x + y) \approx f \ x + f \ y) \\ &\rightarrow (\forall y \rightarrow \llbracket [f] \ y \rrbracket \rho s \approx f(\llbracket y \rrbracket \rho s)) \\ &\rightarrow (f \ 0\# \approx 0\#) \\ &\rightarrow \forall \ x s \\ &\rightarrow \Sigma? \llbracket \text{poly-map } [f] \ x s \rrbracket (\rho, \rho s) \\ &\quad \approx f(\Sigma \llbracket x s \rrbracket (\rho, \rho s)) \end{aligned}$$

Now, each operation only has to be proven up to the semantics defined above. Crucially, optimisations like the sparse encoding *respect* these semantics, so we only have to change our proof in one place: the definition of `poly-mapR`.

4.6 Benchmarks

As expected, the sparse implementation exhibits a significant speedup in type-checking over the dense implementation. Fig. 5a shows time taken to type check a proof that $(x_1 + x_2 + x_3 + x_4 + x_5)^d$ is equal to its expanded form. The sparse representation is clearly faster overall, with a factor of 7.5 speedup for $d = 8$.

Fig. 5b demonstrates where some of the speedup might come from. The expression x^d is represented very differently in the two implementations: in the sparse encoding, it's a single-element list containing a tuple of 1 and d ; in the dense encoding, however, it's a list of length d , with a single 1 at the end. Since the complexities of arithmetic operations are bounded by the length of the list, this explains the difference in cost of the addition operations.




Fig. 5. Time (in seconds) to prove each expression is equal to its expanded form ($n = 5$ for each). Benchmarks performed on Agda version 2.6-0fa9b13, with the Agda standard library at commit-3bd3334a9552490e396f73f96812105a27e5917b, on a 2016 MacBook Pro, with a 2.9 GHz Intel Core i7 and 16 GB of RAM.

Fig. 5c demonstrates perhaps a more common use-case, with a mix of high and low powers and some constants. The sparse representation’s advantage is even more apparent here, with an 30-factor speedup at $d = 8$.

The dense solver does exhibit a small lead (roughly 2-3 seconds) on very simple expressions, possibly caused by the overhead of the sparse solver’s more complex implementation. 3 seconds is quite small in the context of Agda type checking (the standard library, for instance, takes several minutes to type check), so we feel this slight loss is more than made up for by the several-minute gains. Furthermore, we have not been able to find a case where the dense solver is significantly faster than the sparse. Nonetheless, if a user really wants to use the dense solver, the other components described here are entirely modular, and can work with any underlying solver which uses the reflexive technique.

5 PEDAGOGICAL SOLUTIONS

One of the most widely-used and successful computer algebra systems, especially among non-programmers, is Wolfram|Alpha [Wolfram Research, Inc. 2019]. It can generate “pedagogical” (step-by-step) solutions to maths problems [The Development Team 2009]. For instance, given the input $x^2 + 5x + 6 = 0$, it will give the following output:

$$\begin{aligned}
 x^2 + 5x + 6 &= 0 \\
 (x + 2)(x + 3) &= 0 \\
 x + 2 = 0 \text{ or } x + 3 &= 0 \\
 x = -2 \text{ or } x + 2 &= 0 \\
 x = -2 \text{ or } x &= -3
 \end{aligned}$$

These tools can be invaluable for students learning basic mathematics. Unfortunately, much of the software capable of generating usable solutions is proprietary (including Wolfram Alpha), and little information is available as to their implementation techniques. [Lioubartsev 2016] is perhaps the best current work on the topic, but even so very little work exists in the way of the theoretical basis for pedagogical solutions.

Lioubartsev [2016] reformulates the problem as one of *path-finding*. The left-hand-side and right-hand-side of the equation are vertices in a graph, where the edges are single steps to rewrite an expression to an equivalent form. A^* is used to search.

This approach suffers from a huge search space: every vertex will have an edge for almost every one of the ring axioms, and as such a good heuristic is essential. Unfortunately, what this should be is not clear: Lioubartsev [2016] uses a measure of the “simplicity” of an expression.

So, with an eye to using our solver instead of A^* , we can notice that paths in undirected graphs form a perfectly reasonable equivalence relation: transitivity is the concatenation of paths, reflexivity is the empty path, and symmetry is *reversing* a path. Equivalence classes, in this analogy, are connected components of the graph.

More practically speaking, we implement these “paths” as lists, where the elements of the list are elementary ring axioms. When we want to display a step-by-step solution, we simply print out each element of the list in turn, interspersed with the states of the expression (the vertices in the graph).

```

module Trace {a}
  {A : Set a} where

  data _..._ (x : A) : A → Set a where
    [] : x ... x
    _::_ : ∀ {y z}
      → String
      → y ... z
      → x ... z

  refl : Reflexive _..._
  refl = []


  trans : Transitive _..._
  trans [] ys = ys
  trans (x :: xs) ys = x :: trans xs ys

  sym : Symmetric _..._
  sym = go []
  where
    go : ∀ {x y z} → y ... z → y ... x → x ... z
    go xs [] = xs
    go xs (y :: ys) =
      go ("sym(" ++ y ++ ")") :: xs) ys

```

If we stopped there, however, the solver would output incredibly verbose “solutions”: far too verbose to be human-readable. Instead, we must apply a number of heuristics to cut down on the solution length:

- (1) First, we filter out “uninteresting” steps. These are steps which are obvious to a human, like associativity, or evaluation of closed terms. When a step is divided over two sides of an operator, it is deemed “interesting” if either side is interesting.
- (2) Next, we remove any “step, reverse-step” chains. Since we’re converting expressions to normal form, the path may be hourglass-shaped. Fig. 6 is the output from our solver without this heuristic applied. The problem is that both expressions hit a common form early on, at $x + y + 3$.



Nonetheless, the proof will soldier on,
giving superfluous steps.
These sections are what we want to de-
tect, and remove.

After applying those heuristics, our solver
outputs the following for the lemma in Fig. 1b:

```

x + y * 1 + 3
  = { eval }
x + y + 3
  = { +-comm(x, y + 3) }
y + 3 + x
  = { +-comm(y, 3) }
3 + y + x
  = { eval }
2 + 1 + y + x

```

Figuring out good heuristics and path compression techniques seems to deserve further examination.

6 RELATED WORK

In dependently-typed programming languages, the state-of-the-art solver for polynomial equalities (over commutative rings) was originally presented in Grégoire and Mahboubi [2005], and is used in Coq’s ring solver. This work improved on the already existing solver [Coq Development Team 2002] in both efficiency and flexibility. In both the old and improved solvers, a reflexive technique is used to automate the construction of the proof obligation (as described in Boutin [1997]).

Agda [Norell and Chapman 2008] is a dependently-typed programming language based on Martin-Löf’s Intuitionistic Type Theory [Martin-Löf 1980]. Its standard library [Danielsson 2018] currently contains a ring solver which is similar in flexibility to Coq’s ring, but doesn’t support the reflection-based interface, and is less efficient to the one presented here.

In Slama and Brady [2017], an implementation of an automated solver for the dependently-typed language Idris [Brady 2013] is described. The solver is implemented with a “correct-by-construction” approach, in contrast to Grégoire and Mahboubi [2005]. The solver is defined over *noncommutative* rings, meaning that it is more general (can work with more types) but less powerful (meaning it can prove fewer identities). It provides a reflection-based interface, but internally uses a dense representation.

Reflection and metaprogramming are relatively recent additions to Agda, but form an important part of the interfaces to automated proof procedures. Reflection in dependent types in general is explored in Christiansen [2015], and specific to Agda in van der Walt [2012].

Formalisation of mathematics in general is an ongoing project. Wiedijk [2018] tracks how much of “The 100 Greatest Theorems” [Kahl 2004] have so far been formalised (at time of writing, the number stands at 93). DoCon [Meshveliani 2018] is a notable Agda library in this regard: it contains many tools for basic maths, and implementations of several CAS algorithms. Its implementation is described in Meshveliani [2013]. Cheng et al. [2018] describes the manipulation of polynomials in both Haskell and Agda.

Finally, the study of *pedagogical* CASs which provide step-by-step solutions is explored in Lioubartsev [2016]. One of the most well-known such system is Wolfram Alpha [Wolfram Research, Inc. 2019], which has step-by-step solutions [The Development Team 2009].

REFERENCES

- G Allais. 2011. Deciding Presburger Arithmetic Using Reflection. (May 2011). <https://gallais.github.io/pdf/presburger10.pdf>
- Samuel Boutin. 1997. Using Reflection to Build Efficient and Certified Decision Procedures. In *Theoretical Aspects of Computer Software (Lecture Notes in Computer Science)*, Martin Abadi and Takayasu Ito (Eds.). Springer Berlin Heidelberg, 515–529.
- Edwin Brady. 2013. Idris, a General-Purpose Dependently Typed Programming Language: Design and Implementation. *Journal of Functional Programming* 23, 05 (Sept. 2013), 552–593. <https://doi.org/10.1017/S095679681300018X>
- Chen-Mou Cheng, Ruey-Lin Hsu, and Shin-Cheng Mu. 2018. Functional Pearl: Folding Polynomials of Polynomials. In *Functional and Logic Programming (Lecture Notes in Computer Science)*. Springer, Cham, 68–83. https://doi.org/10.1007/978-3-319-90686-7_5
- David Raymond Christiansen. 2015. *Practical Reflection and Metaprogramming for Dependent Types*. Ph.D. Dissertation. IT University of Copenhagen. <http://davidchristiansen.dk/david-christiansen-phd.pdf>
- The Coq Development Team. 2002. *The Coq Proof Assistant Reference Manual, Version 7.2*. <http://coq.inria.fr>
- Nils Anders Danielsson. 2018. The Agda Standard Library. <https://agda.github.io/agda-stdlib/README.html>
- Jeremy Gibbons. 2007. Metamorphisms: Streaming Representation-Changers. *Science of Computer Programming* 65, 2 (2007), 108–139. <https://doi.org/10.1016/j.scico.2006.01.006>
- Benjamin Grégoire and Assia Mahboubi. 2005. Proving Equalities in a Commutative Ring Done Right in Coq. In *Theorem Proving in Higher Order Logics (Lecture Notes in Computer Science)*, Vol. 3603. Springer Berlin Heidelberg, Berlin, Heidelberg, 98–113. https://doi.org/10.1007/11541868_7
- Nathan W. Kahl. 2004. The Hundred Greatest Theorems. <http://web.archive.org/web/20080105074243/http://personal.stevens.edu/~nkahl/Top100Theorems.html>
- Pepijn Kokke and Wouter Swierstra. 2015. Auto in Agda. In *Mathematics of Program Construction (Lecture Notes in Computer Science)*, Ralf Hinze and Janis Voigtländer (Eds.). Springer International Publishing, 276–301. <http://www.staff.science.uu.nl/~swier004/publications/2015-mpc.pdf>
- Dmitrij Lioubartsev. 2016. *Constructing a Computer Algebra System Capable of Generating Pedagogical Step-by-Step Solutions*. Ph.D. Dissertation. KTH Royal Institute of Technology, Stockholm, Sweden. <http://www.diva-portal.se/smash/get/diva2:945222/FULLTEXT01.pdf>
- Per Martin-Löf. 1980. *Intuitionistic Type Theory*. Padua. <http://www.cse.chalmers.se/~peterd/papers/MartinL%00f6f1984.pdf>
- Sergei D Meshveliani. 2013. *Dependent Types for an Adequate Programming of Algebra*. Technical Report. Program Systems Institute of Russian Academy of sciences, Pereslavl-Zalessky, Russia. 15 pages. <http://ceur-ws.org/Vol-1010/paper-05.pdf>
- Sergei D. Meshveliani. 2018. DoCon-A a Provable Algebraic Domain Constructor. <http://www.botik.ru/pub/local/Mechveliani/docon-A/2.02/>
- Shin-Cheng Mu, Hsiang-Shang Ko, and Patrik Jansson. 2009. Algebra of Programming in Agda: Dependent Types for Relational Program Derivation. *Journal of Functional Programming* 19, 5 (Sept. 2009), 545–579. <https://doi.org/10.1017/S0956796809007345>
- Ulf Norell. 2018. Agda-Prelude: Programming Library for Agda. <https://github.com/UlfNorell/agda-prelude>
- Ulf Norell and James Chapman. 2008. Dependently Typed Programming in Agda. (2008), 41.
- Franck Slama and Edwin Brady. 2017. Automatically Proving Equivalence by Type-Safe Reflection. In *Intelligent Computer Mathematics*, Herman Geuvers, Matthew England, Osman Hasan, Florian Rabe, and Olaf Teschke (Eds.). Vol. 10383. Springer International Publishing, Cham, 40–55. https://doi.org/10.1007/978-3-319-62075-6_4
- The Coq Development Team. 2018. The Coq Proof Assistant, Version 8.8.0. <https://doi.org/10.5281/zenodo.1219885>
- The Development Team. 2009. Step-by-Step Math. <http://blog.wolframalpha.com/2009/12/01/step-by-step-math/>
- Paul van der Walt and Wouter Swierstra. 2013. Engineering Proof by Reflection in Agda. In *Implementation and Application of Functional Languages*, Ralf Hinze (Ed.). Vol. 8241. Springer Berlin Heidelberg, Berlin, Heidelberg, 157–173. https://doi.org/10.1007/978-3-642-41582-1_10
- P. D. van der Walt. 2012. *Reflection in Agda*. Master’s Thesis. Universiteit of Utrecht. <https://dspace.library.uu.nl/handle/1874/256628>
- Freek Wiedijk. 2018. Formalizing 100 Theorems. <http://www.cs.ru.nl/~freek/100/>
- Wolfram Research, Inc. 2019. Wolfram|Alpha. Wolfram Research, Inc.. <https://www.wolframalpha.com/>

A APPENDIX

Text of appendix ...